

MATHEMATIK für die ASTRONOMIE

KULTURBUND DER DDR

Zentrale Kommission Astronomie und Raumfahrt – Arbeitskreis „Numerische Astronomie“

Veröffentlichung 3

Leipzig 1987

Liebe Freunde des Arbeitskreises!

Wir haben im Jahre 1986 versucht, durch Versenden von Fragebögen Informationen ueber Aktivitaeten, Wuensche und Vorstellungen von den an unserer Arbeit interessierten Sternfreunden zu erhalten.

Dieses Vorhaben ist uns nur teilweise gelungen. Schwierigkeiten im technischen Ablauf und lueckenhafte Beantwortung der Fragen geben keinen zusammenfassenden Ueberblick.

Deutlich wurde das Interesse an Problemen der numerischen Astronomie, aber nicht die Vorstellung ueber eigene Beitraege dazu.

Die Beschaeftigung mit der numerischen Astronomie setzt solide Kenntnisse der theoretischen Astronomie sowie ihrer mathematischen und physikalischen Grundlagen voraus.

Da nicht in genuegendem Umfang Literatur fuer jeden Sternfreund zur Verfuegung steht, werden wir diese Veroeffentlichung nutzen und in stark verkuerzter Form die notwendigen Grundlagen vermitteln. Dabei stehen das astronomische Phaenomen und der mathematische Algorithmus im Vordergrund und nur selten die rechentechnische Umsetzung.

Wir hoffen, auf diesem Weg vielen Interessenten Anregungen zur eigenen Beschaeftigung zu geben. Unterschiedliche Schwierigkeitsgrade in den theoretischen Anforderungen und verschiedene Genauigkeitsansprueche werden in den Veroeffentlichungen deutlich gemacht.

Wir sind immer an Meinungen zu unserem Material und an selbst gewonnenen Erfahrungen damit interessiert. Auch eigene Beitraege und Anregungen sind gefragt.

Das Programm unserer Veroeffentlichungen f. d. naechsten Jahre umfasst die folgenden Themen (nicht in der Reihenfolge des Erscheinens)

- *Kulmination Auf- u. Untergangszeiten f. Sterne u. and. Himmelsobjekte
- *Ephemeridenrechnung fuer Planeten und Kometen (mehrere Ausgaben)
- *Mondphasen, Sonnen- und Mondfinsternisse
- *Theorie der Finsterniszyklen
- *Globaler Verlauf einer Sonnenfinsternis
- *Stern- und Planetenbedeckungen durch den Mond
- *Mathematische Methoden (Interpolation, Approximation, num. Integration und Differentiation) - Aufgaben der Stoerungsrechnung
- *Demonstrationsprogramm zur astronomischen Phaenomenologie
- *Sonnen- u. Mondephemeride (Diskussion der erreichbaren Genauigkeit)
- *Bahnbestimmung aus Beobachtungen (Kreis-, Ellipsen- u. Parabelbahn)
- *Reduktionsrechnungen

Der Arbeitskreis plant fuer

Sonnabend, den 10.10.1987
von 10.00 - 16.00 Uhr im
Astron. Zentr. Schkeuditz

ein Seminar, zu dem alle Interessenten eingeladen sind. Fuer auswaertige Mitglieder im Kulturbund erfolgt die Fahrgeldrueckerstattung nach Reichsbahntarif.

Im Mittelpunkt des Seminars stehen Probleme der Ephemeridenrechnung. Fuer den Nachmittag sind Kurzvortraege der Mitglieder vorgesehen. Eine Teilnahme sollte schriftlich angezeigt werden, ebenso die Bereitschaft zu einem Kurzvortrag mit Thema und Inhaltsangabe.

gez. Lothar Ehrenberg
Leiter des AK

=====
Aus Zuschriften entnehme ich, dass eine der haeufigsten Schwierigkeiten bei der rechentechnischen Umsetzung numerischer Probleme die Rechengenauigkeit ist.

Diese Veroeffentlichung soll bei der Loesung solcher Schwierigkeiten helfen.

Die meisten Interessenten von "MATHEMATIK f.d. ASTRONOMIE" besitzen einen programmierbaren Taschenrechner, einen Heimcomputer oder haben anderweitig Zugang zu Rechentechnik. Ausserdem beziehen Schul- und Volkssternwarten, die mit Kleincomputern ausgeruestet sind, diese Veroeffentlichung. Ich bitte deshalb um das Verstaendnis derjenigen, die ueber keine programmierbare Rechentechnik verfuegen.

Die beste Moeglichkeit, Fehlern zu entgehen, ist das schrittweise Abarbeiten eines Programms mit der Ausgabe aller Zwischenergebnisse. So wird man feststellen, dass mitunter $A \uparrow 2$ nicht unbedingt das gleiche Ergebnis bringt, wie $A * A$.

Beispiel:

Die Berechnung von Polynomen: $Y_0 + Y_1 * T + Y_2 * T \uparrow 2 + Y_3 * T \uparrow 3$

Die bessere Loesung: $T * (T * (T * Y_3 + Y_2) + Y_1) + Y_0$

Eine weitere Fehlerquelle ist die Tatsache, dass verschiedene Rechner den Integerwert im negativen Zahlenbereich nicht mathematisch korekt liefern.

$INT(-3.5)$ muss -4 und nicht -3 ergeben. (Der Integerwert ist immer die naechstkleinere ganze Zahl)

Etwas schwieriger wird es, wenn ganz einfach die Stellenzahl des Rechners zu gering ist. Auch hier hilft es zunaechst, wenn man das Programm schrittweise abarbeitet. Moeglicherweise ergeben sich waehrend des Rechenganges Winkel mit sehr grossen Absolutbetrageen, wodurch mitunter Stellen hinter dem Komma "verschenkt" werden. Hier ist es hilfreich, Winkel immer wieder auf einen Betrag unterhalb von 360 zu reduzieren.

Es ist auch moeglich, durch geschicktes Umstellen eines Programms eine hoehere Genauigkeit zu erzielen, als sie der Rechner urspruenglich bietet. Ein Beispiel dafuer ist ein Programm zur Errechnung des Julianischen bzw. Buergerlichen Datums fuer einen Kleincomputer aus der KC-Reihe, welche ueber eine Genauigkeit von nur sieben Stellen verfuegen. Der Programmator ist Lothar Ehrenberg, Leiter des AK Numerische Astronomie.

```

1980 REM **** BUERGERLICHES - JULIANISCHES DATUM ****
1990 :
2000 J1=JA:M1=MO
2010 IF MO>2 GOTO 2030
2020 J1=JA-1:M1=MO+12
2030 IF JA>1582 GOTO 2080
2040 IF JA<1582 THEN LET BB=0:GOTO 2100
2050 IF MO > 10 GOTO 2080
2060 IF MO < 10 THEN LET BB=0:GOTO 2100
2070 IF TA < 15 THEN LET BB=0:GOTO 2100
2080 AA=INT(J1/100)
2090 BB=2-AA+INT(AA/4)
2100 JD=INT(365.25*J1)+BB+TA
2110 JD=JD+INT(30.6001*(M1+1))
2120 :
2980 REM **** JULIANISCHES - BUERGERLICHES DATUM ****
2990 :
3000 ZZ=JD:AA=ZZ
3010 IF ZZ<578166 GOTO 3040
3020 A1=INT(ZZ/36524.2-4.00340)
3030 AA=ZZ+A1+1-INT(A1/4)
3040 BB=AA
3050 CC=4715+INT(BB/365.25+0.665708)
3060 DD=INT(365.25*(CC-4715)+0.75)
3070 EE=INT((366+BB-DD)/30.6001)
3080 TA=366+BB-DD-INT(30.6001*EE)
3090 IF EE<13.5 THEN LET MO=EE-1
3100 IF EE>13.5 THEN LET MO=EE-13
3110 IF MO< 2.5 THEN LET JA=CC-4715
3120 IF MO> 2.5 THEN LET JA=CC-4716

```

Eine andere Moeglichkeit, zumindest die Grundrechenarten mit groesserer Genauigkeit abzuwickeln, ist ein BASIC-Unterprogramm, das auf fast allen Rechnertypen laufen kann.

Die einzige Aufbereitung vor der Abarbeitung in diesem Unterprogramm besteht darin, alle Variablen in zwei Teilvariablen zu zerlegen: Den integeren und den fraktionellen Teil. Weiterhin muss ein Operationsstring zugewiesen werden. (+; *; /; Q oder K) - Subtraktion ist Addition einer negativen Zahl. (Vorzeichen fuer beide Teilvariablen)

Nach Abarbeitung im Unterprogramm besteht das Ergebnis wieder aus zwei Teilvariablen, deren Summe das eigentliche Ergebnis ist.

Dieses Programm ist sicher ausbaubar, jedoch die Loesung mathematischer Funktionen in einem BASIC-Unterprogramm waere zu aufwaendig. Ausserdem wuerde die Rechenzeit zu stark ansteigen.

```

10 REM ----- TESTPROGRAMM -----
20 :
30 INPUT"ERSTER OPERAND (VOR KOMMA)";Q1
40 INPUT"ERSTER OPERAND (NACH KOMMA)";Q2
50 INPUT"ZWEITER OPERAND (VOR KOMMA)";W1
60 INPUT"ZWEITER OPERAND (NACH KOMMA)";W2
70 INPUT"OPERATION ( + * / Q K )";OP$:GOSUB6040
80 PRINT"ERGEBNIS",E1;ABS(E2):Q1=E1:Q2=E2:GOTO50
90 END
6000 REM ***** DOPPELTE GENAUGIGKEIT *****
6010 :
6020 A1=ABS(Q1):A2=ABS(Q2):B1=ABS(W1):B2=ABS(W2):RETURN
6030 E2=A2+B2:AA=INT(E2):E2=E2-AA:E1=A1+B1+AA:RETURN
6040 IFOP$="+"THEN6130
6050 IFOP$="*"THEN6250
6060 IFOP$="/"THEN6340
6070 IFOP$="Q"THEN6410
6080 IFOP$="K"THEN6450
6090 PRINT:PRINT"KEIN OP$":STOP
6100 :
6110 REM ----- ADDITION/SUBTRAKTION -----
6120 :
6130 Q3=Q1+Q2:W3=W1+W2:S1=SGN(Q3):S2=SGN(W3):IFS1=0ANDS2=0THEN6210
6140 GOSUB6020:IFS1=S2THENGOSUB6030:E1=E1*S1:E2=E2*S1:RETURN
6150 IFA1>B1THENSG=S1:E2=A2-B2:A3=INT(E2):E2=(E2-A3)*SG
6160 IFA1>B1THENE1=(A1-B1+A3)*SG:RETURN
6170 IFA1<B1THENSG=S2:E2=B2-A2:A3=INT(E2):E2=(E2-A3)*SG
6180 IFA1<B1THENE1=(B1-A1+A3)*SG:RETURN
6190 IFA2>B2THENE1=0:SG=S1:E2=(A2-B2)*SG:RETURN
6200 IFA2<B2THENE1=0:SG=S2:E2=(B2-A2)*SG:RETURN
6210 E1=0:E2=0:RETURN
6220 :
6230 REM ----- MULTIPLIKATION -----
6240 :
6250 QQ=Q1+Q2:WW=W1+W2:S1=SGN(QQ):S2=SGN(WW):SG=-1:IFS1=S2THENSG=1
6260 IFS1=0ORS2=0THENE1=0:E2=0:RETURN
6270 GOSUB6020
6280 Z1=A1*B1:Z2=A2*B2:XX=A1*B2:YY=A2*B1
6290 A1=INT(XX):A2=XX-A1:B1=INT(YY):B2=YY-B1:GOSUB6030:A1=Z1:A2=Z2
6300 B1=E1:B2=E2:GOSUB6030:E1=E1*SG:E2=E2*SG:RETURN
6310 :
6320 REM ----- DIVISION -----
6330 :
6340 QQ=Q1+Q2:S1=SGN(QQ):S2=SGN(W1):SG=-1:IFS1=S2THENSG=1
6350 IFS1=0THENE1=0:E2=0:RETURN
6360 GOSUB6020:XX=A1/B1:YY=A2/B1:A1=INT(XX):A2=XX-A1:B1=INT(YY)
6370 B2=YY-B1:GOSUB6030:E1=E1*SG:E2=E2*SG:RETURN
6380 :
6390 REM ----- QUADRIEREN -----
6400 :
6410 A1=Q1*Q1:A2=Q2*Q2:XX=2*Q1*Q2:B1=INT(XX):B2=XX-B1:GOSUB6030
6420 RETURN
6430 REM ----- KUBIK -----
6440 :
6450 SG=SGN(Q1+Q2):Q1=ABS(Q1):Q2=ABS(Q2):GOSUB6410
6460 A1=E1:A2=E2:B1=Q1:B2=Q2:GOSUB6280:RETURN

```

Eine weitere Moeglichkeit der Erhoehung der Rechengenauigkeit stellt das in unserem Arbeitskreis entwickelte Programm "High Precision Calculator" ("HPC") dar.

Es handelt sich um ein etwa 4,5 KByte langes Maschinenprogramm fuer den Mikroprozessor Z80/U880, mit dem bis zu 75-stellige Rechnungen im Gleitpunktformat ausgefuehrt werden koennen.

"HPC" laesst sich an alle Mikrorechner auf der Basis des Mikroprozessors Z80/U880 anpassen, wenn mindestens 8 KByte RAM zur Verfuegung stehen.

Das Programm "HPC" stellt **kein BASIC** dar, vielmehr wird in diesem Programm mit einer eigenen Programmiersprache gearbeitet, die in ihrem Befehlsumfang mit einem programmierbaren Taschenrechner, der mit umgekehrter polnischen Notation arbeitet, verglichen werden kann.

In "HPC" gibt es Befehle fuer arithmetische Operationen, Standardfunktionen, bedingte und unbedingte Spruenge, Zaehlschleifen, Stapelbefehle und Befehle zur Arbeit mit adressierbaren Speichern.

Damit sich das Programm an jeden Computer anpassen laesst, der mit dem entsprechenden Mikroprozessor arbeitet, geschieht die Uebergabe aller Daten in vom Anwender festlegbaren RAM-Bereichen als ASCII-Zeichenkette.

Die praktische Arbeit mit dem Programm sei an einem kleinen Beispiel erlaeutert:

Es soll der Funktionswert $y = \sinh(x) = (e^x - e^{-x})/2$ eines Eingabewertes x berechnet werden.

Dazu muss mit einem Texteditor des Anwenders (Texteditor ist **nicht** in "HPC" enthalten) folgendes Programm als Text erstellt werden:

Befehl	Erlaeuterung	Calculatorstapel
INP	Eingabe x	x
DUP	x doppeln	x x
EXP	e^x	e^x x
SWAP	tauschen	x e^x
NEG	negieren	-x e^x
EXP		e^{-x} e^x
-		e^x $-e^{-x}$
.2E1	Konstante "2"	2 e^x $-e^{-x}$
/		$(e^x - e^{-x})/2$
PRT	Ausgabe	%
END		

Dieser Text muss als ASCII-Zeichenkette auf den Speicherbereich geladen werden, wo ihn "HPC" erwartet. Danach wird der Compiler aufgerufen. Er teilt dem Benutzer in einem anderen Speicherbereich, dem Ein/Ausgabepuffer mit, ob die Uebersetzung fehlerfrei ist.

Nach erfolgreichem Compilieren kann der Wert x vom Anwender in den Ein/Ausgabepuffer geschrieben werden und der Calculator gestartet werden. Er kehrt ins aufrufende Programm zurueck und der Anwender kann das Ergebnis als Zeichenkette aus dem Ein/Ausgabepuffer entnehmen.

Es ist empfehlenswert, dass die Nutzung dieses Programms nur von Interessenten erwogen wird, die schon ueber eine gewisse Erfahrung mit Computern und insbesondere im Umgang mit Maschinenprogrammen haben.

Esexistieren schon fertig angepasste Versionen fuer Z 9001 (mit BASIC-ROM) und Sinclair ZX81/64K.

Ernsthafte Interessenten wenden sich bitte direkt an den Programmautor

Uwe Pilz Lazarusstrasse 23 Leipzig 7024

=====

Arbeitskreis "Numerische Astronomie"
im Astronomischen Zentrum Schkeuditz
7144 Schkeuditz PSF 29
Leiter: Dipl.- Phys. Lothar Ehrenberg

=====

Autor: Guenter Bethge Kreuzstrasse 3B Leipzig 7010

mit Beitraegen von Lothar Ehrenberg und Uwe Pilz

=====